New Features in SAS Version 8

by Elvira Agrón

CIT/TASC/AST
National Institutes of Health
March 7, 2001

Seminar Description

This seminar will review some new features available with version 8 of the SAS System. It is geared towards experienced SAS users. Among the topics to be covered are:

- New SAS windowing environment
- Configuring the SAS application
- Enhanced editor
- Naming conventions
- Length of strings, variables and data set names
- Importing/exporting files
- Output delivery system
- Review new procedures and procedure options

SAS Windowing Environment



Explorer Window

With the Explorer window you can manage SAS files and create shortcuts to non-SAS files. You can create, open, copy, move, view and delete SAS files.

Managing SAS Libraries

To manage SAS files double-click **Libraries**.

To create a new SAS library



Select **File** → **New**

(or click on the toolbutton)

Enter a name for the library (libref) and the path for the library. You may want to select "Enable at startup" to automatically define this library every time you start SAS.

Using **New Library** is equivalent to submitting a **LIBNAME** statement.

Explorer Window

Managing SAS Files

Double-click on the library **In** to view a list of all the SAS files saved in that library. Right-clicking on any SAS file will bring up a popup menu with various options such as: open, view columns, export, copy, delete and properties.



Properties Window



To move up one level from the **Explorer** window click on the toolbutton 

To remove the **Explorer** window for all sessions modify the configuration file SASV8.CFG (in your SAS folder). Replace the line "-dmsexp" with "-dms".

Results Window

With the Results window you can view, save, print and manage the listing and HTML output.

File Shortcuts

You can define file shortcuts in the SAS Explorer window. For example, you can create shortcuts for SAS programs and raw data. You can use the shortcuts in several SAS statements like %INCLUDE and INFILE.

Double-click on the **File Shortcuts** icon then right-click in the **Active File Shortcuts** window then select **New File Shortcut**. Enter the shortcut and filename and press **OK**.

Note: Under version 8.0 shortcuts are only active during the current SAS session.



My Favorite Folders

You can define your most frequently used folders by selecting the **View** menu then **My Favorite Folders**. Right-click on the area named "Contents of Favorite Folders" then select **New Favorite Folder**. Enter a name for the folder, type the full pathname of the folder then press **OK**.



Now you can easily find, open and submit files from your frequently used folders directly from SAS.

You can open **My Favorites Folder** automatically at startup in the **Explorer** window by clicking the **Explorer** then **Tools** → **Options** → **Explorer**. Select **Initialization** from the first drop-down list then select **Favorites** and press **Add**.

Change Working Folder

To change the current working folder permanently add the system option -**sasinitialfolder** in the configuration file SASV8.CFG (in your SAS folder), resave the file and restart SAS.



The Preferences Window

To customize the SAS workspace select **Tools → Options → Preferences.** Make any changes you wish to make then press **OK**.





The Enhanced Editor

The new **Enhanced Editor** is used to edit and submit SAS programs. It provides many new useful features such as:

- Multiple instances
- Color-coding for SAS keywords
- Expandable and collapsible sections
- Recordable keyboard macros
- Abbreviations
- Submitted program remains in the Enhanced Editor window
- Asterisk after program name indicates the file has not been saved



Note: F5 is not associated with the Enhanced Editor. F5 opens the old Program Editor. Pressing F8 from the Output window puts you back in the last Enhanced Editor window submitted.

Enhanced Editor Options

To modify the Enhanced Editor select **Tools** → **Options** → **Enhanced Editor** from any Enhanced Editor window.



Enhanced Editor Keys

To view the enhanced editor keyboard shortcuts select **Tools** → **Options** → **Enhanced Editor Keys** from any Enhanced Editor window.



Examples:

Collapse all folding blocks

Alt+Ctrl+Number pad –

Expand all folding blocks

Alt+Ctrl+Number pad +

Go to line

Ctrl + G

Move to matching brace/parenthesis

Ctrl + [or Ctrl +]

Move cursor to first visible line

Alt + Up

Move cursor to last visible line

Alt + Down

Comment the selection

Ctrl + /

Uncomment the selection

Shf + Ctrl + /

Recording Keyboard Macros

You can create a keyboard macro to issue a group of commands that you use frequently.

To define a keyboard macro select **Tools → Keyboard Macros → Record New Macro** from any Enhanced Editor window, or press **Alt+Shift+R**.

Here we will enter a PROC MEANS step. When done select **Tools → Keyboard Macros → Stop Recording** or press **Alt+Shift+R**. In the **Save Keyboard Macro** window enter a macro name and a description then press **Assign keys**. In the **Assign keys** window press the shortcut keys of your preference, press **Assign** then **OK**.

To insert PROC MEANS in any program we just need to press Alt+Shift+M.

Managing Keyboard Macros

To edit, rename or delete keyboard macros select **Tools → Keyboard Macros → Macros.**

Abbreviations

To create abbreviations for frequently used text select **Tools → Add Abbreviation**. Enter the abbreviation and the text to insert for the abbreviation then press **OK**.



Anytime you enter the abbreviation in the Enhanced Editor window SAS will give you a chance to substitute the abbreviation for the text it represents. Press **Enter** to replace the abbreviation.

Importing Files

To import a file select **File → Import Data**. Select the appropriate data source from the **Select import type** window then press **Next**.

Here we'll import an Excel 2000 file. Enter the source file name then press **Next**. Depending on your data you may need to press **Options** before moving forward.



Importing Files

Choose the destination of the new SAS file then press **Finish** to import the file or **Next** if you want to save the program SAS creates behind the scenes.



SAS will convert the file to a SAS data set.

Exporting Files

To export a file select **File → Export Data**. Select a library and member (SAS data set) then press **Next**.

Here we'll export the file to an Access 2000 table. Select the data source type then press **Next**.





Exporting Files

Enter the name of the Access database and press **OK**. Then enter a name for the new Access table and press **Finish** to export the file. Pressing **Next** gives you the chance of saving the statements SAS creates behind the scenes.





Length of SAS Names and Values

The following SAS names can now be up to 32 characters long:

- data set variables
- data set names
- macro variables
- macros
- arrays

These still have a limit of 8:

- librefs
- filerefs
- Formats

This still has a limit of 7:

- informats

Variable and data set labels can now be up to 256 characters long.

The maximum length for character data is now 32767.

Data set variable names can be of mixed case but are internally capitalized for processing (i.e. BalanceAmount is the same as balanceamount).

VALIDVARNAME= Option

The VALIDVARNAME= option controls the type of SAS variable names that can be created and processed. It can take the following values:

V6	variable names must be saved in uppercase, contain no more than 8 characters and begin with a letter or underscore followed by letters, underscores or numbers.
V7	variable names can be up to 32 characters long, preserve mixed case and begin with a letter or underscore followed by letters, underscores or numbers.
UPCASE	same as V7 but can only be uppercase
ANY	allows any characters to appear as a valid SAS column name. It must be of the form 'varname'n: 'visit date'n, 'Balance-Amount'n. ANY is experimental in version 8.

Example:

```
options validvarname=v6;
```

Specifying Data Set Names

In Version 8 you can create and use a SAS data set using its pathname. You don't have to use a LIBNAME statement.

Example:

```
data 'c:\awards\data\awards99';
  input . . . ;
run;

proc freq data='c:\awards\data\awards99';
run;
```

Library Engines

When you submit a LIBNAME statement without an engine SAS will assign an engine of:

- V8 if there are no SAS files in the folder or at least one version 8 file
- V6 if all the SAS files in the folder are version 6.10, 6.11 or 6.12 files
- V604 if all the SAS files in the folder are version 6.03 or 6.04 files

When you create or use many types of SAS data sets in the same program it is best to explicitly specify the engine. For example, the first statement below can be used to access the 6.12 data set AWARDS. The second statement points to the SAS data sets AWARDS98 and NEWAWARDS.

```
libname in1 v6 'c:\awards\data_mixed';
libname in2 v8 'c:\awards\data_mixed';
```



ODBC Engine

In version 8 of SAS you can use the ODBC engine in a LIBNAME statement to point to databases. This includes: Microsoft Access, ORACLE and SQL Server databases.

In this example we use the ODBC engine to point to a the Microsoft Access database. Before running the LIBNAME statement you must define the DSN (data source name) using the ODBC Manager from the Control Panels window in Windows.

```
libname in odbc dsn='Awards Data';
proc freq data=in.awards9697;
  table state;
run;
```

Converting SAS Files

- Version 8 can read and write version 6 SAS data sets.
- It can read but not write version 6 SAS format libraries.
- To convert SAS files from version 6 to 8 you can use PROC COPY or the SAS Explorer.

Example 1: Converting all the SAS files from a version 6 library to 8:

```
libname old v6 'c:\awards\data_v6';
libname new v8 'c:\awards\data';
proc copy in=old out=new;
run;
```

Example 2: Converting selected members from a version 6 library to 8:

```
libname old v6 'c:\awards\data_v6';
libname new v8 'c:\awards\data';
proc copy in=old out=new;
  select formats awards99;
run;
```

Example 3: Converting all the SAS files from a version 8 library to 6:

```
libname old v6 'c:\awards\data_v6';
libname new v8 'c:\awards\data';
proc copy in=new out=old;
run;
```

Creating SAS Data with Viewtable

You can enter data directly into SAS with the **VIEWTABLE** window. The result will be a SAS data set. To open this window select **Tools → Table Editor**.

To assign names and attributes to the variables right-click on the variable name and select **Column Attributes**. After making changes press **Apply**.

To save the data select **File → Save As**. Select a library and enter a member name (i.e. a SAS data set name).

Generation Data Sets

In Version 8 you can keep multiple copies of a SAS data set by using the GENMAX= option. The multiple copies represent versions of the same data set, which is archived each time it is replace. To access any version of the data set use the GENNUM= option.

```
data test(genmax=3);
  input id $ n;
datalines;
A 10
B 20
;
data test;
  set test;
  if id='A' then n=50;
run;

data test;
  set test;
  if id='B' then n=500;
run;
```

```
*Prints the current version;
proc print data=test;
run;

*Prints the version one generation
back from the current version;
proc print data=test(gennum=-1);
run;

*Prints the version two generations
back from the current version;
proc print data=test(gennum=-2);
run;
```

SAS Help

You can get SAS help in various ways:

1. Help → SAS System Help
2. SAS OnlineDoc (**http://statsoft.nih.gov/pubs/onldoc.htm**). For security reasons, your mainframe (OS/390) **initials** and password are required.
3. Pressing F1 on any SAS keyword in the Enhanced Editor window

The SAS OnlineDoc can be accessed directly from the SAS System by:

1. modifying the configuration file **SASV8.CFG** that is saved in your SAS folder. Modify the line that starts with **–DOCLOC** so it points to the SAS OnlineDoc URL above. Then you can access it by selecting **Help → Books and Training → SAS OnlineDoc**.
2. or, adding a tool button that is associated with the SAS OnlineDoc URL:
   a. select **Tools → Customize** then select the **Customize** tab.
   b. from the **Add Tool** button select **Blank Tool**. 
   c. in the **Command** text box enter the URL. The **Help Text** and **Tip Text** are optional.
   d. select the **Change Icon** button, select a button then press **OK**. 
   e. press **Yes** after the question "Do you want to save the changes?"

SAS OnlineTutor

The SAS OnlineTutor provides 35 lessons that you can take with any standard Web browser. It is designed for Version 8.

The URL for the tutor is: **http://statsoft2.nih.gov:8000/sastutor**

For security reasons, your mainframe (OS/390) **initials** and password are required.

To start the tutor:

1. If you want to practice what you learn, first select **Help** → **Books and Training** → **SAS OnlineTutor** in SAS. This sets up the data that you can use for practice. Alternatively, you can issue the following command from SAS's command bar: **af c=sashelp.oltutor.datasets.scl**. (The command bar is the white box at the upper left-hand corner of the SAS application.)
2. Open your browser and point to the website specified above.

You can assign the URL above to a tool button to easily start the SAS OnlineTutor. Follow the steps outlined in part 2 of the previous page.

For more information about the SAS OnlineTutor visit our website: **http://statsoft.nih.gov**.

Output Delivery System

A new feature in version 7 that is available for all procedures is the output delivery system, or ODS. It allows you to save SAS output in many formats such as HTML, Postscript, SAS data sets and RTF (experimental in 8.0, development in 8.1).

For example, to save the results of PROC REPORT in an HTML file, include an ODS HTML statement with the BODY= option before invoking PROC REPORT. After the RUN statement add the ODS HTML statement with the CLOSE option.

```
ods html body='c:\awards\res_awards.html';
title 'NIH Research Awards Granted to Medical Schools';
title2 'District of Columbia, Maryland and Virginia';
proc report data=in.awards nowd headskip headline;
 column year stname resn resamt;
 define year / 'Year' group;
 define stname / 'State' group;
 define resn / 'N' f=comma5.;
 define resamt / 'Total' f=dollar12.;
 break after year / skip summarize ol suppress;
run;
ods html close;
```

Hint: To easily export tables to Excel you can read the HTML file with Excel and save the file as an Excel table.

Output Delivery System



NIH Research Awards Granted to Medical Schools
District of Columbia, Maryland and Virginia

Year	State	N	Total
1996	DIST OF COL	174	$44,120,947
	MARYLAND	781	$228,938,871
	VIRGINIA	337	$78,893,975
		1,292	*$351,953,793*
1997	DIST OF COL	182	$47,236,672
	MARYLAND	817	$238,185,108
	VIRGINIA	357	$85,447,478
		1,356	*$370,869,258*

Output Delivery System

To save an output table in a SAS data set you must first know the name of the object. You can obtain it in various ways:

- for any SAS/STAT procedure, from the procedure documentation,
- using the ODS TRACE statement

The following program writes a trace record for the tables created by UNIVARIATE in the LOG:

```
ods trace on;
proc univariate data=in.awards;
 var resamt;
run;
ods trace off;
```

Partial LOG:

```
Output Added:
-------------
Name:        Moments
Label:       Moments
Template:    base.univariate.Moments
Path:        Univariate.RESAMT.Moments
-------------

Output Added:
-------------
Name:        BasicMeasures
Label:       Basic Measures of Location and Variability
Template:    base.univariate.Measures
Path:        Univariate.RESAMT.BasicMeasures
-------------

Output Added:
-------------
Name:        TestsForLocation
Label:       Tests For Location
Template:    base.univariate.Location
Path:        Univariate.RESAMT.TestsForLocation
-------------

Output Added:
-------------
Name:        Quantiles
Label:       Quantiles
Template:    base.univariate.Quantiles
```

Output Delivery System

To save any table use the ODS OUTPUT statement followed by the table name or path and the SAS data set name. For example, these statements save the MOMENTS table in a temporary SAS data set called RESMOM.

```
ods output moments=resmom;
proc univariate data=in.awards;
 var resamt;
run;

*Or;
ods output resamt.moments=resmom;
proc univariate data=in.awards;
 var resamt;
run;
```



To suppress the listing in the OUTPUT window specify the ODS LISTING CLOSE statement:

```
ods output moments=resmom;
ods listing close;
proc univariate data=in.awards;
 var resamt;
run;
ods listing;
```

Some New Procedures

Base SAS

EXPORT	reads data from a SAS data set and writes it to an external data source
IMPORT	reads data from an external data source and writes it to a SAS data set

SAS/STAT

SURVEYSELECT	provides a variety of methods for selecting probability-based random samples
SURVEYMEANS	produces estimates of survey population means and totals from sample survey data
SURVEYREG	performs regression analysis for sample survey data
LOESS	implements a nonparametric method for estimating regression surfaces
NLMIXED	fits nonlinear mixed models
BOXPLOT	creates side-by-side box-and-whisker plots

Quantile Statistics

The procedures MEANS, SUMMARY and TABULATE now produce quantile statistics by using the following keywords:

MEDIAN | P50	Q3 | P75
P1	P90
P5	P95
P10	P99
Q1 | P25	QRANGE

The procedure TABULATE also includes the following keywords used to calculate row, column, page and report percentages:

ROWPCTN	ROWPCTSUM
COLPCTN	COLPCTSUM
PAGEPCTN	PAGEPCTSUM
REPPCTN	REPPCTSUM

Multilabels in PROC FORMAT

The FORMAT procedure offers the new option MULTILABEL that lets you specify multiple values for a given range, or for overlapping ranges. You can only use multilabels with the procedures MEANS, SUMMARY and TABULATE .

```
proc format;
value res (multilabel)
  low-42519003='Below average'
  42519003-high='Above average'
  low-25000000='Group A'
  25000000-50000000='Group B'
  50000000-high='Group C';
run;

proc tabulate data=in.awards format=8.;
 class resamt / mlf;
 class year;
 table resamt all, year all*(n colpctn) / rts=15;
 format resamt res.;
run;
```

	YEAR			
	1996	1997	All	
	N	N	N	ColPctN
RESAMT				
Above average	2	3	5	29
Below average	6	6	12	71
Group A	3	4	7	41
Group B	4	3	7	41
Group C	1	2	3	18
All	8	9	17	100

Web Graphics Driver

SAS/GRAPH includes the following web drivers: GIF, HTML and WEBFRAME.

Example 1: Creating a GIF file

```
filename gout1 'c:\awards\graphs\graph1.gif';
goptions gsfname=gout1 dev=gif;
proc gchart data=in.awards;
  vbar stname / group=year sumvar=resn;
run; quit;
```

Example 2: Creating a GIF file and an HTML file

```
filename gout2 'c:\awards\graphs';
goptions gsfname=gout2 dev=html;
proc gchart data=in.awards;
  vbar stname / group=year sumvar=resn;
run; quit;
```

Example 3: Creating thumbnail-size graphs

```
filename gout3 'c:\awards\graphs';
goptions gsfname=gout3 dev=webframe;
proc gchart data=in.awards;
  vbar stname / sumvar=resn;
  by year;
run;quit;
```

Using ODS for Graphics

You can also use the Output Delivery System to create HTML and GIF files.

```
goptions dev=gif373;
ods listing close;    *so it doesn't create SASGRAPH.GIF;
ods html path='c:\awards\graphs' body='myfile.htm';
proc gchart data=in.awards;
  vbar stname / subgroup=year sumvar=resn;
run;
ods html close;
ods listing;
```

Some advantages of using the ODS to create web output are:

- you can name the body file
- combine graphics and non-graphics output in the same web page
- create drill-down graphs

Drill-down Graphics

You can create drill-down graphics using the HTML= option, the HTML_LEGEND= option or both. The following procedures support drill-down graphics: GCHART, GMAP and GPLOT.

To create a drill-down graph:

1. Add a new column to the data set that contains the HREF link to the file containing detail information.
2. Specify a GIF driver in the GOPTIONS statement.
3. Use the ODS HTML statement to specify a path for your files and the body file for the graph.
4. Create the graph using the new data set. In the appropriate statement set HTML= to the new column created in step 1.
5. Create each file containing the detail information including an ODS HTML statement to create each individual file.
6. Close the HTML destination.

Drill-down Graphics

Example:

```
*Step 1;
data awards2;
  set in.awards;
  if stname='DIST OF COL' then link='href=dc.htm';
  else if stname='MARYLAND' then link='href=md.htm';
  else if stname='VIRGINIA' then link='href=va.htm';
run;

*Step 2;
goptions reset=all dev=gif570;

*Step 3;
ods listing close;
ods html path='c:\awards\graphs' body='research.htm';

*Step 4;
title 'Number of Research Awards';
proc gchart data=awards2;
  vbar stname / sumvar=resn html=link;
run; quit;
```

Drill-down Graphics

Example:

```
*Step 5;
ods html body='dc.htm';
proc print data=in.awards(where=(stname='DIST OF COL')) noobs;
 title 'WASHINGTON, D.C.';
 var univname resn resamt;
 footnote '<a href="research.htm">Graph</a>';
run;

ods html body='md.htm';
proc print data=in.awards(where=(stname='MARYLAND')) noobs;
 title 'MARYLAND';
 var univname resn resamt;
 footnote '<a href="research.htm">Graph</a>';
run;

ods html body='va.htm';
proc print data=in.awards(where=(stname='VIRGINIA')) noobs;
 title 'VIRGINIA';
 var univname resn resamt;
 footnote '<a href="research.htm">Graph</a>';
run;

*Step 6;
ods html close;
ods listing;
```

Drill-down Graphics



Clicking on the bar **DIST OF COL** opens the listing for DC.



Clicking on **Graph** takes you back to the graph.

3D Charts

The GCHART procedure now includes three new statements to produce 3D charts: HBAR3D, VBAR3D and PIE3D. The following program produces the 3D vertical bar chart shown below.

```
proc gchart data=in.awards;
  vbar3d stname / subgroup=year sumvar=resn;
run;
```



Suggested Reading

1. Help → Getting Started with SAS Software
2. What's New for SAS Software Products, Version 8
3. Chapter 2 of the Windows Companion, "Interacting with the SAS System under Windows"
4. The Complete Guide to the SAS Output Delivery System
5. Chapter 5 of the SAS/GRAPH Guide, "Bringing SAS/GRAPH Output to the Web"
6. Individual chapters of the new procedures

Items 2 through 6 are available in SAS OnlineDoc.